Calculating Hamming Distance Using Assembly Programming

Murali Indukuri

February 21, 2025

0.1 Introduction

In this lab, a program to calculate the hamming distance of two strings using the assembly programming language is presented. It works by using the XOR operator to find bits that differ between the two strings and then sums the differing bits. The support up to a kilobyte of text for each input string and can report up to 9,999 differences.

Hamming distance is a measure of the number of bits that differ between two strings. For instance, the bit-strings 01010101 and 10101010 have a hamming distance of 8. All printable characters are encoded using ASCII codes, which are 7-bit numbers. By definition, the XOR operator outputs a high bit only if the inputs differ; therefore, the operator can be used to compute the hamming distance by isolating differing bits. To compute the sum of all the different bits, the result of the XOR operation can be continuously bit shifted to the left or right. If a bit is set high, it will also set the carry flag when it is shifted left or right. Adding the value of the carry flag to a counter will calculate the number of different bits in a given character for both strings. Therefore, an additional loop must be used to loop over all characters. One of the requirements of this program was to only compare the first n characters of both strings where n is the length of the shortest input string. Therefore, the outer loop will break once it finds a newline character in one of the strings. Then, the value of the counter is converted to decimal using integer division. This final value is stored in a single 32-bit register, limiting it to four characters(9,999 differences). The value is printed to the console. Future work may store the final error value on the stack or use 64-bit registers to increase the number of errors reported.

0.2 Full Code

The following shows the full assembly program. Note: the program relies heavily on Linux System Calls, and will not work on any other operating system. Furthermore, it will only work on x86 machines.

```
1  section .bss
2    str1: resb 1025
3    str2: resb 1025
4
5  section .data
6    ;prompt for first string
7    str1_prmpt: db "Enter the first string: "
8    str1_prmpt_len: equ $-str1_prmpt
9    str2_prmpt: db "Enter the second string: "
10   str2_prmpt_len: equ $-str2_prmpt
11   output_msg: db "Hamming Distance: "
12   output_msg_len: equ $-output_msg
13
14 section .text
15   global _start
16
17 _start:
18   ; Read input
19   ; call prompt print
20   push str1_prmpt
```

```
21  push str1_prmpt_len
22  call print
23
24  ;get str1 input
25  mov eax, 3
26  mov ebx, 0
27  mov ecx, str1
28  mov edx, 1025
29  int 0x80
30
31  ; call prompt print
32  push str2_prmpt
33  push str2_prmpt_len
34  call print
35  ; get str2 input
36  mov eax, 3
37  mov ebx, 0
38  mov ecx, str2
39  mov edx, 1025
40  int 0x80
41
42  ; Used as counter (DO NOT CHANGE)
43  xor edx, edx ; use edx for counter
44  xor edi, edi ; zero out pointer
45
46  char_loop:
47    xor eax, eax
48    mov al, byte [str1+edi]
49    ;and eax, 0x000000FF  ;Mask out three bytes
50
51    xor ebx, ebx
52    mov bl, byte [str2+edi]
53    ;and ebx, 0x000000FF
54
55    ; Exit if EOL
56    cmp al, 0x0A
57    je report_count
58    cmp bl, 0x0A
59    je report_count
60
61    xor al, bl ; compare char
62    ; if different by 1, eax looks like:
63    ; 000000000000001
64
65    ; nested loop
66    push edi
67    xor edi, edi
68    bit_cmp_loop:
69      shr al, 1 ; bitshift to carry
70      jnc loop_end
```

```
71    add edx, 1
72        loop_end:
73
74        add edi, 1
75        cmp edi, 8
76        jl bit_cmp_loop
77        pop edi
78
79      ; increment count
80      add edi, 1
81      cmp edi, 1025
82      jl char_loop
83
84      report_count:
85      push edx
86      ; Print output msg
87      push output_msg
88      push output_msg_len
89      call print
90      ; Print hamming distance
91      ; note that edx is still pushed
92      call int_print
93
94      ; Format newline
95      push 0x0A ; newline char
96      push esp
97      push 1
98      call print
99      call exit
100
101
102 print:
103   ; create call frame
104   push ebp ; Remember that this is 32bit(4 bytes)
105   mov ebp, esp
106   ; Function body
107   mov eax, 4
108   mov ebx, 1
109   mov ecx, [ebp+12]
110   mov edx, [ebp+8]
111   int 0x80
112   ; dump stack frame
113   pop ebp
114   ret 8
115
116 int_print:
117   push ebp
118   mov ebp, esp
119
120   mov eax, [ebp+8] ; get int
```

```
  cmp eax, 0
  jne checked_zero
    test:
    mov eax, 0x30303030 ; 0000
    push eax
    push esp
    push 4
    call print
    pop eax
    pop ebp
    ret 4
  checked_zero:

  xor edi, edi
  xor ecx, ecx
  loop_divide:
    ; Divide
    xor edx, edx
    mov ebx, 10
    div ebx
    add edx, 48
    shrd ecx, edx, 8
    add edi, 1
    cmp edi, 4
    jl loop_divide
    ; Output
    ; flip register
    mov ebx, ecx
    xor ecx, ecx
    or ch, bl
    or cl, bh
    ror ecx, 16
    ror ebx, 16
    or ch, bl
    or cl, bh
    ; Finally print
    push ecx
    push esp
    push 4
    call print
    pop eax
  pop ebp
  ret 4

exit:
  mov eax, 1
  int 0x80
```

0.3 Sample Output

The following shows the output of the program for selected inputs:

```
murali@murali-Inspiron-16-Plus-7630:~/Documents/Code/ASM$ ./build-asm.sh hamming-distance
Enter the first string: foo
Enter the second string: bar
Hamming Distance: 0008
```

Figure 1: The binary string for foo is 011001100110111101101111. For bar, it is 011000100110000101110010. They differ by 8 bits.

```
murali@murali-Inspiron-16-Plus-7630:~/Documents/Code/ASM$ ./build-asm.sh hamming-distance
Enter the first string: this is a test
Enter the second string: of the emergency broadcast system
Hamming Distance: 0038
```

Figure 2: This shows how the program only considers characters up to the length of the shortest string.



Figure 3: This shows the ability of the program to handle up to a kilobyte(1024 characters) of text. the "#" character differs from the "\" character by 7 bits. As expected, the distance for 1024 characters differing in 7 bits is 7168.

```
murali@murali-Inspiron-16-Plus-7630:~/Documents/Code/ASM$ ./build-asm.sh hamming-distance
Enter the first string: @
Enter the second string: A
Hamming Distance: 0001
```

Figure 4: The "A" character differs from the "@" character by exactly one bit. This was often used for troubleshooting.